

OPEN SOURCE MEDIA FRAMEWORK

Developer's Guide

© 2010 Adobe Systems Incorporated. All rights reserved.

Open Source Media Framework Developer's Guide

This guide is protected under copyright law, furnished for informational use only, is subject to change without notice, and should not be construed as a commitment by Adobe Systems Incorporated. Adobe Systems Incorporated assumes no responsibility or liability for any errors or inaccuracies that may appear in the informational content contained in this guide.

This guide is licensed for use under the terms of the Creative Commons Attribution Non-Commercial 3.0 License. This License allows users to copy, distribute, and transmit the guide for noncommercial purposes only so long as (1) proper attribution to Adobe is given as the owner of the guide; and (2) any reuse or distribution of the guide contains a notice that use of the guide is governed by these terms. The best way to provide notice is to include the following link. To view a copy of this license, visit <http://creativecommons.org/licenses/by-nc-sa/3.0/>

Adobe, the Adobe logo, Adobe AIR, ActionScript, AIR, Flash, Flash Access, Flash Builder, Flex, Flex Builder, and Omniture are either registered trademarks or trademarks of Adobe Systems Incorporated in the United States and/or other countries. All other trademarks are the property of their respective owners.

Adobe Systems Incorporated, 345 Park Avenue, San Jose, California 95110, USA.

Notice to U.S. Government End Users: The Software and Documentation are "Commercial Items," as that term is defined at 48 C.F.R. §2.101, consisting of "Commercial Computer Software" and "Commercial Computer Software Documentation," as such terms are used in 48 C.F.R. §12.212 or 48 C.F.R. §227.7202, as applicable. Consistent with 48 C.F.R. §12.212 or 48 C.F.R. §§227.7202-1 through 227.7202-4, as applicable, the Commercial Computer Software and Commercial Computer Software Documentation are being licensed to U.S. Government end users (a) only as Commercial Items and (b) with only those rights as are granted to all other end users pursuant to the terms and conditions herein. Unpublished-rights reserved under the copyright laws of the United States. Adobe agrees to comply with all applicable equal opportunity laws including, if appropriate, the provisions of Executive Order 11246, as amended, Section 402 of the Vietnam Era Veterans Readjustment Assistance Act of 1974 (38 USC 4212), and Section 503 of the Rehabilitation Act of 1973, as amended, and the regulations at 41 CFR Parts 60-1 through 60-60, 60-250, and 60-741. The affirmative action clause and regulations contained in the preceding sentence shall be incorporated by reference.

Contents

Chapter 1: OSMF Overview

Features	1
Development environment	1

Chapter 2: Basic Playback

Media elements and resources	4
Media factories	6
Media players and media containers	7
Media player sprites and a simpler player	8

Chapter 3: Interaction

Simple user interaction	10
-------------------------------	----

Chapter 4: Delivering Content

Delivery methods	13
Optimizing delivery and enhancing playback	16

Chapter 5: Enhancing Content

Proxy elements	19
Metadata	20

Chapter 6: Advanced Topics

Logging	25
Traits	27
External HTML control with JavaScript	28
Plug-ins	30

Chapter 1: OSMF Overview

Open Source Media Framework (OSMF) is a pure ActionScript® 3.0 framework that gives developers complete flexibility and control in creating their own rich media experiences. OSMF is a free collection of open-source components that simplify media player creation for the Adobe® Flash® Platform.

The OSMF production-ready code base saves you time and effort, so you have more time for the features with the greatest impact for your customers. OSMF also has an open and extensible architecture that allows you to integrate third-party services quickly and easily.

The framework, documentation, and latest updates are on the OSMF website: www.osmf.org. For further support, see www.adobe.com/support/osmf.

Features

OSMF provides the following features:

- Support for both standard and advanced delivery methods, including progressive download; plain RTMP and HTTP streaming; RTMP and HTTP Dynamic Streaming; and live streaming with RTMP, HTTP, and RTMFP multicast.
- Automatic content protection with Flash® Access™ 2.0.
- Simple integration of third-party plug-in services such as CDNs, advertising, and analytics. The flexible architecture supplies the option of compiling plug-ins statically or loading them dynamically, so plug-in providers can perform immediate upgrades and versioning.
- Quality of service (QoS) features, including dynamic (multi-bitrate) streaming and Stream Reconnect.
- Support for all Flash® Player 10.0 media formats, including:
 - Streaming video (FLV, F4V, MP4, MPEG-4, MP4, M4V, F4F, 3GPP)
 - Progressive audio (mp3)
 - Progressive video (FLV, F4V, MP4, MP4V-ES, M4V, 3GPP, 3GPP2, QuickTime)
 - Images (PNG, GIF, JPG)
 - SWF files
 - Limited support for streaming audio (mp3, AAC, Speex, Nellymoser)

Development environment

To use OSMF, plug-in and player developers must have a minimum of Flash Player 10.0 or later installed as part of their development environment. End users must also have Flash Player 10.0 or later installed.

Flash Player 10.1 is required for full access to all OSMF 1.5 features, including:

- HTTP Streaming (also requires Flash® Media Server 3.8 or greater)
- Stream Reconnect (also requires Flash Media Server 3.5.3 or greater)

- RTMFP multicast (also requires Flash Media Server 4.0 or greater)

OSMF is implemented using ActionScript 3.0, without any Flex® APIs or Flash authoring components, so it's not dependent on a particular development platform.

To use OSMF in the development environment of your choice:

- Download the OSMF source zip file.
- Extract the OSMF.swc file from the source zip.
- Add it to your library folder or path. (The details vary depending on which development environment and platform you are using.)

Specific directions for Adobe development environments follow.

Flex Builder 3 setup

Flex® Builder™ 3 users may need to update their development environment to use OSMF 1.5.

- 1 Open Flex Builder and choose File > New > Flex (ActionScript) Project. Select Flex Builder, and then Flex Project. Name the project. Select web application as application type. Click Finish.
- 2 Select the project. Choose File > Properties. In the Properties window, select Flex Compiler.
- 3 Verify the Flex SDK version. The latest SDK from Flex 3 is 3.5. To download the latest Flex 3 SDK, go to www.adobe.com/go/flex3_sdk.
- 4 OSMF supports Flash Player version 10.1 as default. You may need to update playerglobal.swc in your Flex SDK. To do so:
 - a Download playerglobal.swc from http://download.macromedia.com/pub/labs/flashplayer10/flashplayer10_globalswc.zip.
 - b Replace the old playerglobal.swc with the new one. It should be located in .../Flex Builder3/sdks/[your sdk version]/frameworks/libs/player/10/
 - c To install or update Flash player, go to <http://get.adobe.com/flashplayer>.
- 5 Do one of the following to use the OSMF source:
 - Copy OSMF.swc on to your local machine. In the Properties window of your project, select the Flex Build Path ?> Library Path tab. Select Add SWC. Add OSMF and click Finish.
 - Or, copy the OSMF source to your local machine. Open Flex Builder and choose Import > Existing projects into workspace. Click Browse to go to where the OSMF package is copied. Select OSMF and Click Finish. In your project Properties window, select Flex Build Path and select the Library path tab. Click Add Project and browse to the folder that contains the OSMF package.
- 6 While OSMF itself is independent of any Flex SDK version, OSMF and the sample apps that ship with it are created with Flex SDK 4.1 or above. Therefore, to avoid compiler issues, you may need to do the following:
 - a Remove -show-invalid-css-property-warnings=false from compiler option. In the project properties, select "Flex (Library) Compiler". In Additional compiler arguments, remove the compiler option.
 - b Add htmlPlayerVersion="10.0.0" to .actionscriptProperties in OSMF lib and any sample apps provided by OSMF.

Flash Builder 4 setup

To avoid potential compatibility issues, Flash® Builder™ 4 users should change their default SDK version to Flex 4.1 and not select the “Use Flex 3 compatibility mode” checkbox.

- 1 Open “Flash Builder 4” and choose File > New > Flex (ActionScript) Project. Name the project. Select web application as application type. Click Finish.
- 2 Select the project. Choose File > Properties. In the Properties window, select Flex (ActionScript) Compiler.
- 3 Verify the Flex SDK version is Flex 4.1. To download the Flex SDK 4.1, go to <http://opensource.adobe.com/wiki/display/flexsdk/Download+Flex+4>.
- 4 Set Require Flash Player version to 10.1.0. To install or update Flash player, go to <http://get.adobe.com/flashplayer>.
- 5 Do one of the following to use the OSMF source:
 - Copy OSMF.swc on to your local machine. In your Properties window of your project, select the Flex (ActionScript) Build Path > Library Path tab. Select Add SWC. Add OSMF and click Finish.
 - Or, copy the OSMF source to your local machine. Open Flash Builder and choose Import > Existing projects into workspace. Next. Click browse to go to where the OSMF package is copied. Select OSMF and Click Finish. In the project Properties window of your project, select Flex Build Path and select the Library path tab. Click Add Project and browse to the folder that contains OSMF package.

Flex 4.0 SDK setup

If you are using the Flex® 4.0 SDK, you may need to update the installed version of Flash Player.

- 1 Download the current playerglobal.swc from http://download.macromedia.com/pub/labs/flashplayer10/flashplayer10_globalswc.zip. Copy it to your local directory: /Adobe Flash Builder 4/sdks/4.0.0/frameworks/libs/player/.
- 2 If you have created a 10.1 folder to copy the playerglobal.swc (.../Adobe Flash Builder 4/sdks/4.0.0/frameworks/libs/player/10.1), do the following. In your project properties, under Adobe Flash Player options, select “Use a specific version” and choose 10.1.0.
- 3 To remove the older version of OSMF that shipped with Flex 4 SDK, select File > Properties > Flex Build Path > Library path, select osmf.swc under Flex 4.0 SDK on the Library Path panel, and click the Remove button.

Adobe Flash Professional CS 5 setup

Follow these steps to use OSMF with Adobe® Flash® Professional CS5.

- 1 Create your Flash project.
- 2 Select File > Publish settings. Go to the Flash tab. Set the player version to Flash Player 10 and ActionScript 3.0.
- 3 Click Settings, move to Library path tab and click the swc file icon.
- 4 Add OSMF.swc.
- 5 If you choose to publish an HTML file that detects the version of Flash Player, detect version 10.0.0 or later.

Chapter 2: Basic Playback

Open Source Media Framework (OSMF) provides basic building blocks that you can quickly use to create your media player and begin playback:

- “[Media elements and resources](#)” on page 4 and “[Media factories](#)” on page 6 present the initial concepts and steps for how your player handles content with OSMF.
- “[Media players and media containers](#)” on page 7 describes how to use the OSMF classes that play and display content.
- “[Media player sprites and a simpler player](#)” on page 8 presents the simplest method of starting basic playback.

Media elements and resources

Any type of content that an OSMF media player plays is considered to be a type of media element. Examples of media elements include videos, sound files, images, or SWF content. A media element can also be a grouping of media that is played as if it is a single media file.

`MediaElement` is the base class that represents every category of media in the framework. All media types included in the framework extend `MediaElement`. Custom media types introduced by developers must also extend `MediaElement`. Using `MediaElement`, OSMF players can be notified of and respond to events within the display list hierarchy for Flash Player applications.

To construct a media element, you supply a media resource, specifying the content to play. Typically, you use a URL or file path as the defining data for a media resource. However, a media resource can also be something more complex, such as an array of streams or a plug-in application. `URLResource` is the type most commonly used to instantiate media elements.

Media element types

OSMF defines many types of standard media elements. Among these, the core elements represent basic media types, including:

- `AudioElement`. This type supports both streaming and progressive delivery formats and can play mp3 or AAC files. It handles mp3 files over HTTP, as well as audio-only streams from Flash Media Server.
- `ImageElement`. You can use this type to load and present PNG, GIF, or JPG images.
- `VideoElement` and `LightweightVideoElement`. `LightweightVideoElement` supports progressive and simple RTMP streaming only. It is the parent class for `VideoElement`, which supports all OSMF video delivery modes.
- `SWFElement`. Use this element to play SWF (Flash movie) files.

Each of the core media element types is a child of the `LoadableElementBase` class, a direct subclass of `MediaElement`. Including `LoadableElementBase`, the direct subclasses of `MediaElement` are:

- `LoadableElementBase`, which provides base functionality for media elements that require loading before they are played. It is not typically instantiated directly but is the base class for the core media element types.
- `CompositeElement` enables functional groupings of core elements. You do not typically create instances of `CompositeElement` directly, but instead use its subclasses, `ParallelElement` and `SerialElement`. See “[Creating a composite element](#)” on page 5 for more information and a code sample.

- `HTML`Element represents a piece of media external to the Flash SWF file that is within an HTML region. It serves as a bridge to external JavaScript APIs and is discussed in “[External HTML control with JavaScript](#)” on page 28.
- `Proxy`Element is a media element that wraps another media element for the purpose of controlling access to the wrapped element. `Proxy`Element and its subclass `Duration`Element are discussed in “[Proxy elements](#)” on page 19.
- `Beacon`Element is a media element that contains the URL to retrieve when a play request (HTTP GET) is received. It is useful for media analytics/reporting.

Creating a composite element

A composite element describes a single media experience made up of individual parts. Here are a few examples of media compositions:

- 10-minute video stream + midroll ad + additional 10-minute video stream + postroll ad
- 20-minute video stream + companion banner SWF ads shown next to the video
- Two videos at the same time, as in picture-in-picture or multiple camera angles

`Composite`Element defines this type of media element, but typically you do not create instances of this type directly. Instead, you create media compositions based on its two subclasses:

- `Parallel`Element is a media composition whose elements are presented in parallel (concurrently).
- `Serial`Element is a media composition whose elements are presented serially (in sequence).

However many individual elements make up a media composition, the overall composition operates as a single, unified media element. For example, if `Serial`Element encapsulates a series of videos, the `Serial`Element object behaves as if it's a single video element, but one that plays several videos in sequence. Or, if a `Parallel`Element object encapsulates an image and a piece of audio, the object behaves as if it's a single media element with the audio characteristics of the audio file and the display characteristics of the image.

Note: *If files of different types are being loaded in a `Parallel`Element, it is possible that a file of one type (for example, an image) may complete loading before other members of the element (such as a progressively downloaded video).*

Because a `Composite`Element object maintains a list of `Media`Element children, any of which can be `Composite`Element objects themselves, a media composition can be expressed as a tree structure.

Note: *Media compositions that include 4 or more parallel elements can perform poorly in some browser environments.*

The following code excerpt shows the basic construction of a composite element.


```
private function createMediaElement():MediaElement
{
    var serialElement:SerialElement = new SerialElement();
    // First child is a progressive video.
    serialElement.addChild
        ( new VideoElement
            ( new URLResource(http://myserver.com/assets/testprogressive.mov)
            )
        );
    // Second child is a Flash movie. DurationElement causes it to play for 3 seconds.
    serialElement.addChild
        ( new DurationElement
            ( 3, new SWFElement
                ( new URLResource(http://myserver.com/assets/testSWF.swf)
                )
            )
        );
    // Third child is a streaming video.
    serialElement.addChild
        ( new VideoElement
            ( new URLResource(http://myserver.com/assets/teststream.mp4)
            )
        );
    return serialElement;
}
```

Media factories

The simplest way to turn a media resource into an element of the appropriate type is to use a media factory. The `DefaultMediaFactory` class takes a media resource as input and translates it into the correct type of `MediaElement`. `DefaultMediaFactory` implements support for most common `MediaElement` subclasses, such as audio, video, image, SWF, and F4M elements.

It is typically best to instantiate a media element indirectly with `DefaultMediaFactory`, rather than doing so directly. By doing so, `DefaultMediaFactory` cannot only create appropriate media elements from URLs, it can also perform convenience services for the content. For example, `DefaultMediaFactory` communicates with proxy plug-ins, so that they know about the element being created and can perform their services for it.

The following code fragment demonstrates how media resources, elements, and factories work together:

```
// Create a new default media factory.
var myFactory:MediaFactory = new DefaultMediaFactory();
// Create the resource to play.
var myResource:URLResource = new URLResource("http://myserver.com/content/test.flv");
// Create a media element using the factory. Specifically, the factory recognizes
// the .flv extension of the resource and creates a VideoElement.
var myElement:MediaElement = myFactory.createMediaElement(myResource);
```

“[Media elements and resources](#)” on page 4 described media resources as the building blocks of media elements. While `DefaultMediaFactory` can typically translate a resource to an element automatically, there are situations where it can't. For example, your player may need to load a file from a URL that has no file extension, such as one served by a PHP script. In this case, you must explicitly provide the file type to `OSMF`; `DefaultMediaFactory` cannot do it for you.

You inform OSMF of the file type by setting the `mediaType` or `mimeType` properties of the media resource. (See `osmf.media.MediaType` for the current list of supported types and their string values.) The following code shows an example of one way to handle this situation:

```
// Create a new image element.
var element:ImageElement = new ImageElement();
// Create a new resource from the URL to a jpg file.
var resource:URLResource = new URLResource("http://myserver.com/content/sample");

// Assign the proper media type to the resource.
resource.mediaType = MediaType.IMAGE;

// Once the resource has its media type, assign the resource to the element.
element.resource = resource;
```

Media players and media containers

The `MediaPlayer` class is the controller class for playing any type of media element. This class roughly corresponds to an actual media player and encapsulates the low-level tasks required to play media. `MediaPlayer` methods and properties map to user interface control actions, such as play, pause, seek, and stop. `MediaPlayer` does not contain any display or “view” functionality.

To play a media element, it must be assigned to a `MediaPlayer` object. Once you do so, `MediaPlayer` automatically starts monitoring the assigned element and tracks when its capabilities become available. The `MediaPlayer` class both automatically loads the assigned media and plays the media once loading completes.

Still, loading and playing are not enough. If you use `MediaPlayer` and `MediaElement`, you can make a simple player application, but no one can see what is playing. From the perspective of the Model-View-Controller software design pattern, the `MediaPlayer` class provides a controller, and the `MediaElement` class is the model. The `MediaContainer` class provides the display or view for your program.

`MediaContainer` is a display object container, descending from the `Sprite` class. `MediaContainer` implements much of the code to display both the content and the interface for your player application. Once you use `MediaContainer` with `MediaPlayer` and `MediaElement`, you have a functional, visible media player.

Note: For more information on display object containers and the display list in the Flash Platform, see `flash.display.DisplayObjectContainer`.

The following sample shows the construction of a simple OSMF media player. This player starts playback automatically, without user interaction. To create an interactive user interface for your player, see [Handling User Interaction](#). To create an even simpler auto-playback media player, see [“Media player sprites and a simpler player”](#) on page 8.

```
package
{
    import flash.display.Sprite;
    import org.osmf.containers.MediaContainer;
    import org.osmf.media.DefaultMediaFactory;
    import org.osmf.media.MediaElement;
    import org.osmf.media.MediaFactory;
    import org.osmf.media.MediaPlayer;
    import org.osmf.media.URLResource;

    [SWF(width="640", height="352")]
    public class HelloWorld extends Sprite
    {
        public function HelloWorld()
        {
            // Create the container class that displays the media.
            var container:MediaContainer = new MediaContainer();
            addChild(container);

            //Create a new DefaultMediaFactory
            var mediaFactory:MediaFactory = new DefaultMediaFactory();

            // Create the resource to play.
            var resource:URLResource = new URLResource("http://myserver.com/test.flv");

            // Create MediaElement using MediaFactory and add it to the container class.
            var mediaElement:MediaElement = mediaFactory.createMediaElement(resource);
            container.addMediaElement(mediaElement);

            // Add MediaElement to a MediaPlayer. Because the MediaPlayer
            // autoPlay property defaults to true, playback begins immediately.
            var mediaPlayer:MediaPlayer = new MediaPlayer();
            mediaPlayer.media = mediaElement;
        }
    }
}
```

Media player sprites and a simpler player

So far we've created a basic media player using `MediaFactory`, `MediaPlayer`, and `MediaContainer`. `MediaPlayerSprite` is an application-level class that combines these classes into one convenience class. It uses `DefaultMediaFactory` to generate the media element and load the content, `MediaContainer` to display the content, and `MediaPlayer` to control the content. (It also allows you to provide a custom media player, container, or factory as parameters, if you want.)

Unlike the `MediaPlayer` class, `MediaPlayerSprite` inherits from `Sprite`, thus allowing an all-in-one component for a player. Creating a media player with `MediaPlayerSprite` is extremely simple and quick, as shown in the sample following.

```
package
{
    import flash.display.Sprite;
    import org.osmf.media.MediaPlayerSprite;
    import org.osmf.media.URLResource;

    [SWF(width="640", height="352")]
    public class HelloWorld extends Sprite
    {
        public function HelloWorld()
        {
            // Create the container class that displays the media.
            var sprite:MediaPlayerSprite = new MediaPlayerSprite();
            addChild(sprite);

            // Assign the resource to play. This generates the appropriate
            // MediaElement and passes it to the MediaPlayer. Because the MediaPlayer
            // autoPlay property defaults to true, playback begins immediately.
            sprite.resource = new URLResource("http://myserver.net/content/test.flv");
        }
    }
}
```

Controlling content scale with MediaPlayerSprite

For visual content, determining how it is sized and scaled is often integral to a successful viewing experience. Because this functionality can be so important, OSMF provides access to scale modes within `MediaPlayerSprite`.

To control the content scale mode, set `MediaPlayerSprite.scaleMode` with one of the following constants:

LETTERBOX	Sets the content width and height as close as possible to that of the container while maintaining the content's original aspect ratio. The player background may appear along the vertical or horizontal edges of the content, if its LETTERBOX size does not exactly match the current dimensions of the container. This is the default scale mode.
NONE	Allows no scaling of content. The content is displayed at its intrinsic size.
STRETCH	Sets the content width and height to that of the container, possibly changing the content's original aspect ratio.
ZOOM	Sets the content to maintain its original aspect ratio, while filling the entire visible area of the container. No background is allowed to appear with the ZOOM setting, so portions of the content may extend beyond the visible bounds of the container and the visible content may appear horizontally or vertically clipped as a result.

Chapter 3: Interaction

“[Basic Playback](#)” on page 4 describes how to create a simple media player. What the player displays, however, is minimal: a window containing content that automatically begins playback. For the user to interact with your player at all, you must implement an interface. Open Source Media Framework provides multiple means of doing so.

You can start by looking at the OSMF player code posted at [Strobe Media Playback](#). The Strobe Media Playback (SMP) project implements a full player interface with event handling. The code is open source and can be edited and reused. For more information, see the [documentation at the SMP wiki](#).

You can also build your own interface for your player. One way to do this is to construct individual interface pieces, add them to the player’s media container, and manage their response to events. An example of this process is presented in “[Simple user interaction](#)” on page 10.

Another way to build a player interface is to implement it as a plug-in. “[Writing a control bar plug-in](#)” on page 31 provides an example implementation.

Simple user interaction

The `Sprite` class is a basic display list building block: a display list node that can display graphics and can also contain children. `Sprite` is an appropriate base class for objects that do not require timelines, such as user interface (UI) components. This example implements a piece of the interface (an overlay Play button) as a `Sprite` object. After the `Sprite` is defined, it is added to the player’s media container and given the ability to respond to events.

Note: When creating a real-world player, more interaction than just an overlay Play button is usually required. Scrub bars, audio buttons, pause buttons, and more are common parts of a player user interface. These controls can also be created using the same approach that is taken for the overlay Play button.

The following function creates the `Sprite` object and constructs the Play button image that visually overlays the player’s content area.

```
private function constructPlayButton():Sprite
{
    var result:Sprite = new Sprite();

    var g:Graphics = result.graphics;
    g.lineStyle(1, 0, 0.5);
    g.beginFill(0xA0A0A0, 0.5);
    g.moveTo(X - SIZE / F, Y - SIZE);
    g.lineTo(X + SIZE / F, Y);
    g.lineTo(X - SIZE / F, Y + SIZE);
    g.lineTo(X - SIZE / F, Y - SIZE);
    g.endFill();

    return result;
}

private static const SIZE:Number = 100;
private static const X:Number = 320;
private static const Y:Number = 180;
private static const F:Number = 1.2;
```

You then add the `Sprite` object to the same container that holds the media element by calling `addChild`:

```
var playButton:Sprite = constructPlayButton();  
addChild(playButton);
```

Next, you add functionality to the button, so that your player can respond when the user clicks Play. You do this by attaching an event listener to the `Sprite` object. This allows the object to receive notification for whatever events you specify. In this case, when a mouse click is detected on the Play button, it causes the media player to begin playback.

```
playButton.addEventListener  
    ( MouseEvent.CLICK,  
      function():void{ mediaPlayer.play(); }  
    );
```

Finally, your player must synchronize the visibility of the Play button with the current state of playback. If playback is paused, the Play button should be visible so the user can start playback up again. But if playback is occurring, the Play button should not be displayed.

To accomplish this, you can add another event listener, this time to your media player class. Your player can then respond to `PlayEvent.PLAY_STATE_CHANGE` by setting the Play button visibility on and off, accordingly.

```
mediaPlayer.addEventListener  
    ( PlayEvent.PLAY_STATE_CHANGE  
      , function():void{ playButton.visible = !mediaPlayer.playing; }  
    );
```

The following sample shows the construction of the Play button overlay in its entirety:

```
package  
{  
    import flash.display.Graphics;  
    import flash.display.Sprite;  
    import flash.events.MouseEvent;  
    import org.osmf.containers.MediaContainer;  
    import org.osmf.events.PlayEvent;  
    import org.osmf.layout.LayoutMetadata;  
    import org.osmf.media.*;  
  
    [SWF(width="640", height="360", backgroundColor="0x000000", frameRate="25")]  
    public class MyOSMFPlayer extends Sprite  
    {  
        public function MyOSMFPlayer()  
        {  
            mediaPlayerSprite = new MediaPlayerSprite();  
            mediaPlayerSprite.mediaPlayer.autoPlay = false;  
            mediaPlayerSprite.resource = new URLResource(VIDEO_URL);  
            // Give the bounds of the MediaPlayerSprite  
            mediaPlayerSprite.width = 640;  
            mediaPlayerSprite.height = 360;  
            addChild(mediaPlayerSprite);  
  
            var playButton:Sprite = constructPlayButton();  
            addChild(playButton);  
  
            playButton.addEventListener  
                (MouseEvent.CLICK,  
                  function():void{ mediaPlayerSprite.mediaPlayer.play(); } );  
  
            mediaPlayerSprite.mediaPlayer.addEventListener
```

Interaction

```
        (PlayEvent.PLAY_STATE_CHANGE, function():void{playButton.visible =
            mediaPlayerSprite.mediaPlayer.playing;} );
    }
}

private var mediaPlayerSprite:MediaPlayerSprite;

private static const SIZE:Number = 100;
private static const X:Number = 320;
private static const Y:Number = 180;
private static const F:Number = 1.2;

private static const VIDEO_URL:String = "http://myserver.net/content/test.flv";
private function constructPlayButton():Sprite
{
    var result:Sprite = new Sprite();

    var g:Graphics = result.graphics;
    g.lineStyle(1, 0, 0.5);
    g.beginFill(0xA0A0A0, 0.5);
    g.moveTo(X - SIZE / F, Y - SIZE);
    g.lineTo(X + SIZE / F, Y);
    g.lineTo(X - SIZE / F, Y + SIZE);
    g.lineTo(X - SIZE / F, Y - SIZE);
    g.endFill();

    return result;
}
}
```

Chapter 4: Delivering Content

The intended audience for your media player may be inside an enterprise network or distributed around the world. Knowing the strengths and limitations of the various “[Delivery methods](#)” on page 13 available to an Open Source Media Framework player is the first step in choosing how you deliver content. OSMF also provides protocol-specific enhancements for your player, as described in “[Optimizing delivery and enhancing playback](#)” on page 16.

Delivery methods

OSMF supports a variety of delivery protocols:

- HTTP, for live and recorded streaming, as well as progressive download
- RTMP, for live and recorded streams

OSMF also supports advanced playback and delivery features for HTTP and RTMP streams. These include digital video recording features and multi-bitrate (“adaptive” or “dynamic”) streaming.

Additionally, new with Flash Media Server 4.0, OSMF also supports:

- RTMFP, to enable multicasting of live streams

About HTTP streaming

HTTP has long been the most basic of content delivery methods. Using HTTP, applications can play video from a standard web server simply by making a request to download a file. This delivery method is known as “progressive” download and has a number of downsides for modern media players. These negatives include greater than necessary bandwidth consumption and the inability to play live content.

OSMF uses Flash Player (version 10.1 or greater) and Flash Media Server (version 3.8 or greater) to provide an alternative to progressive download: HTTP *streaming*. With HTTP streaming, your player can deliver high-quality content more efficiently and with more features, while still using standard HTTP servers. Supported content types include:

- Recorded (does not require Flash Media Server)
- Recorded multi-bitrate (“dynamic”)
- Live
- Live multi-bitrate
- Live with DVR functionality
- Live multi-bitrate with DVR functionality

HTTP streaming also supports each of the previous use cases with Flash Access content protection.

Using HTTP streaming

To use HTTP streaming for a media file, you must supply metadata describing how the media is packaged. You provide this metadata to OSMF in the form of a Flash Media Manifest file (file extension: F4M). Once you create an F4M file, you pass its URL to `DefaultMediaFactory`, which automatically handles loading the stream.

An F4M manifest file is an XML document containing metadata related to a single piece of media. The manifest file can contain references to more than one instance of the media, such as for multiple bitrates. However, the manifest does not encapsulate more than one distinct piece of media and does not function as a playlist.

Manifest information can include the location of the media, DRM authentication information, media bootstrap information, multi-bitrate information (MBR), and more. For a complete description of the F4M format, see [the Flash Media Manifest File Format Specification](#).

For details on the larger HTTP streaming workflow, including methods of creating F4M manifests for live and recorded content, see the [HTTP streaming overview](#) in the document [Using Adobe HTTP Dynamic Streaming](#).

For further information on using HTTP streaming features in your OSMF player, see “[Dynamic Streaming for HTTP and RTMP streams](#)” on page 16 and “[DVR functionality for HTTP and RTMP live streams](#)” on page 18.

About RTMP streaming

Because RTMP was designed as a streaming protocol, it has had many upsides over standard HTTP progressive delivery. These include supporting both recorded and live streams, multi-bitrate streaming, and DVR capabilities. Using Flash Player 10.1 and Flash Media Server 3.5.3, OSMF provides support for new RTMP features, including Stream Reconnect.

Stream Reconnect automatically allows an RTMP stream to continue to play through the buffer when a connection is disrupted. By default, as specified by the `NetLoader.reconnectTimeout` property, OSMF tries to reconnect an RTMP stream for 120 seconds. If the stream is reconnected before the buffer empties, the stream is not interrupted, and buffering restarts unnoticeably. In general, the greater the buffer size, the less chance of stream interruption. “[Managing buffer size for RTMP and HTTP streams](#)” on page 18 provides recommendations on setting the buffer size for RTMP streams.

`DefaultMediaFactory` loads RTMP streams automatically. However, it is important to note that, by convention, RTMP streams should not include a file extension as part of their URL.

Note: *When playing RTMP live streams, your player should not enable Pause button functionality, unless the live stream also has DVR support.*

See the following for more information on using RTMP features in an OSMF player:

- “[Dynamic Streaming for HTTP and RTMP streams](#)” on page 16
- “[DVR functionality for HTTP and RTMP live streams](#)” on page 18

About RTMFP multicasting

HTTP and RTMP are both application-level protocols built on TCP (transmission control protocol). Using TCP is a standard method to obtain reliable data transfers over the web. However, the reliability of TCP comes at the cost of potentially high data latency.

Services that require low latency, as for live interaction with voice over IP or for online games, use the UDP protocol. UDP (user datagram protocol) is an alternative to TCP that delivers data rapidly. UDP is also compatible with packet broadcast (sending to all nodes on a local network) or multicast (sending to all subscribers).

Using UDP multicast, you can build a media player to

- broadcast enterprise-scope information, such as a company meeting
- broadcast video within and beyond your network without a content delivery network (CDN)
- allow internal network clients to participate in a peer-to-peer (P2P) group, to help ensure high-quality video delivery

Using UDP, Adobe has developed the Real-Time Media Flow Protocol (RTMFP), an alternative to TCP-based RTMP and HTTP. To use RTMFP multicast, clients must run an application built for Adobe Flash Player 10.1 or greater or Adobe® AIR®. There are three forms of multicast that RTMFP supports:

Native IP multicast. With IP multicast, a multicast address is associated with a group of interested receivers. The sender sends a single datagram to the multicast address. Then, the intermediary routers copy the message and send the copies to receivers who've registered their interest in data from that sender.

P2P (application-level) multicast. P2P multicast uses Flash Player applications within the multicast group to route and relay data. Because the RTMFP protocol supports groups, this enables an application to segment its users to send messages and data only between members of the group. Application-level multicast provides one-to-many (or a few-to-many) streaming of continuous live video and audio.

Fusion multicast. With Fusion multicast, native IP multicast and P2P/application multicast are combined. Clients can receive content via native IP multicast or fall back to P2P/application multicast if IP multicast is not available.

RTMFP is a managed connection that requires the authorization of a Flash Media Server 4.0 instance to make the introductions. Clients must remain connected to the server to retain the direct connection. Additionally, the server provides a script that handles publish/unpublish requests and republishes the live stream into a target Flash group. For further information on RTMFP, see the [Building Peer-Assisted Networking Applications](#) in the [Flash Media Server 4 Developer's Guide](#).

Note: *Dynamic streaming and DVR functionality (such as pausing or stopping a video) are not supported with RTMFP multicast at this time.*

Using RTMFP multicasting

The OSMF `DefaultMediaFactory` class automatically handles multicast content. You can provide a multicast-enabled resource to `DefaultMediaFactory` in two ways:

- Use the Flash Media Server “Configurator” tool to generate a multicast-enabled F4M manifest file. In Windows, the default location for the Configurator is `C:\Program Files\Adobe\Flash Media Server 4\tools\multicast\configurator`. See “[Using HTTP streaming](#)” on page 13 for more information on F4M files.
- Or, use the `MulticastResource` class to provide a multicast-ready resource for `DefaultMediaFactory` to handle.

When creating and identifying an RTMFP multicast, two important values are the multicast group specification and the stream name.

- The group specification is a string identifying the peer-to-peer group to join, including its name, capabilities, restrictions, and the authorizations of this member.
- The stream name is a string identifying the stream to join. If the supplied string does not refer to an existing application on the server, it is invalid. However, an invalid string only results in a connection not being established; no error is returned.

Note: *If you use the Flash Media Server 4.0.0 Configurator to generate the F4M file, you must manually change the names of two generated values. Change the name `rtmfpGroupspec` to `groupspec`, and the name `rtmfpStreamName` to `multicastStreamName`.*

The following code excerpts compare the ways to instantiate a multicast stream.

Using an F4M file:

```
mediaFactory = new DefaultMediaFactory();  
mediaPlayer.media = mediaFactory.createMediaElement  
(  
    new URLResource("http://example.com/multicast.f4m")  
);
```

Using MulticastResource with DefaultMediaFactory:

```
mediaFactory = new DefaultMediaFactory();  
var resource:MulticastResource = new MulticastResource("rtmfp://server/app");  
resource.groupspec = "AJHJKSFDS";  
resource.streamName = "OPIOPROPIIOPOP";  
mediaPlayer.media = mediaFactory.createMediaElement(resource);
```

Note: For IP multicast, OSMF does not return an error when attempting to connect to a blocked network segment.

Optimizing delivery and enhancing playback

The quality of the user's media experience is what differentiates media players. One important factor for users is delivery quality, specifically how a player responds to bandwidth challenges. "[Dynamic Streaming for HTTP and RTMP streams](#)" on page 16 helps your player adapt to changing network conditions.

Another factor is how much control the user can exercise over their experience. "[DVR functionality for HTTP and RTMP live streams](#)" on page 18 discusses the simple implementation of digital video recorder features with live streaming.

Dynamic Streaming for HTTP and RTMP streams

With Dynamic Streaming, when the player senses a network bandwidth change, it responds by switching playback to a content file with a more appropriate bitrate. Dynamic Streaming requires you to have multiple bitrate (MBR) versions of content for the player to switch among. When you provide MBR content, a network experiencing a temporary reduction in bandwidth does not have to pause playback for the user. Instead, the player seamlessly shifts to using a lower bitrate version of the content that is playing. If the player does not find multi-bitrate versions of content, Dynamic Streaming does not function.

To enable HTTP Dynamic Streaming, you supply OSMF with an F4M file, as described in these [Quick Start tutorials](#).

For RTMP Dynamic Streaming, you can use `MediaPlayerSprite` and `DynamicStreamingResource` as shown in the following sample. `DynamicStreamingResource` contains multiple `DynamicStreamingItem` values, each of which represents a single stream.

```
package
{
    import flash.display.Sprite;
    import flash.display.StageAlign;
    import flash.display.StageScaleMode;
    import org.osmf.elements.VideoElement;
    import org.osmf.media.MediaPlayerSprite;
    import org.osmf.net.DynamicStreamingItem;
    import org.osmf.net.DynamicStreamingResource;

    public class DynamicStreamingExample extends Sprite
    {
        public function DynamicStreamingExample()
        {
            super();

            stage.scaleMode = StageScaleMode.NO_SCALE;
            stage.align = StageAlign.TOP_LEFT;

            var mediaPlayerSprite:MediaPlayerSprite = new MediaPlayerSprite();

            var videoElement:VideoElement = new VideoElement();

            /**
             * By convention, most rtmp URLs do not use file extensions.
             */
            var dynResource:DynamicStreamingResource = new
                DynamicStreamingResource("rtmp://myserver.net/ondemand");

            /**
             * The first two parameters of each DynamicStreamingItem specify
             * stream name and bitrate and are required.
             * The second two parameters of each DynamicStreamingItem are optional;
             * they specify the stream's width and height, in that order.
             */
            dynResource.streamItems = Vector.<DynamicStreamingItem>(
                [ new DynamicStreamingItem("mp4:myserver/content/
                    demo_768x428_24.0fps_408kbps.mp4", 408, 768, 428)
                , new DynamicStreamingItem("mp4:myserver/content/
                    demo_768x428_24.0fps_608kbps.mp4", 608, 768, 428)
                , new DynamicStreamingItem("mp4:myserver/content/
                    demo_1024x522_24.0fps_908kbps.mp4", 908, 1024, 522)
                , new DynamicStreamingItem("mp4:myserver/content/
                    demo_1024x522_24.0fps_1308kbps.mp4", 1308, 1024, 522)
                , new DynamicStreamingItem("mp4:mmyserver/content/
                    demo_1280x720_24.0fps_1708kbps.mp4", 1708, 1280, 720)
                ] );

            videoElement.resource = dynResource;

            addChild(mediaPlayerSprite);
            mediaPlayerSprite.media = videoElement;
        }
    }
}
```

Managing buffer size for RTMP and HTTP streams

OSMF does not set a default size for the RTMP or HTTP stream buffer. You must set `MediaPlayer.bufferTime` to a size appropriate to the needs of your player.

- **For RTMP dynamic streaming/MBR content**, set the buffer size to 8 seconds or greater. This applies to any RTMP dynamic stream, whether live or recorded. For live MBR content, you must set the buffer size to a non-zero value or MBR switching fails.
- **For RTMP streams that are not dynamic/MBR**, set the buffer size to between 2 and 4 seconds.
- **For HTTP Dynamic Streaming**, see the sample [HTTPStreamingNetLoaderWithBufferControl.as](#) in the [OSMFPlayer](#) source repository for an example of managing buffer size.

DVR functionality for HTTP and RTMP live streams

OSMF provides support for digital video recorder (DVR) functions in media players that are playing HTTP or RTMP live streams. Clients can jump back in time while viewing a live stream, pause and resume the live stream, or view the live recording at a later time.

On the server side, providing DVR capability for RTMP streams requires a minimum of Flash Media Server 3.5 and the DVRCast application. To provide DVR capability for HTTP streams, the server must be equipped with Flash Media Server 3.8 or above.

OSMF automatically supports RTMP streams with DVR functionality, as long as the streams are properly identified when the media element is constructed. You must use a `StreamingURLResource` object (rather than a plain `URLResource`) and set the stream type property of the resource to `streamType.DVR`.

For HTTP live streams, enabling DVR features is specified within the F4M file produced by the live packager. For instructions on properly setting up a DVR-enabled F4M file, see [Tutorial: Live HTTP Dynamic Streaming with DVR](#).

Chapter 5: Enhancing Content

You can use OSMF to customize the content your player plays. The following enhancement mechanisms are a good starting point. You may also want to read “[Advanced Topics](#)” on page 25 for more possibilities.

- `ProxyElement` is a powerful `MediaElement` subclass. Learn how “[Proxy elements](#)” on page 19 work, then try out `DurationElement` to create a slide show.
- OSMF also defines a mechanism for the use of metadata. See “[Metadata](#)” on page 20 for information on specific OSMF `Metadata` subclasses to use for content layout and cue points.

Proxy elements


A `ProxyElement` is a class that wraps (proxies) another `MediaElement`, exposing the same API. The purpose of this class is to control access to the wrapped element. The `ProxyElement` class allows layering of functionality and is a core building block for plug-ins, as described in “[Plug-ins](#)” on page 30.

`ProxyElement` is not instantiated directly but rather used as the base class for creating wrappers for specific purposes. `ProxyElement` can be subclassed for any trait or set of traits. The subclass controls access to the wrapped element either by overriding one or more of the wrapped element’s traits or by blocking them. This technique is useful because it allows you to noninvasively modify the behavior of another `MediaElement`.

`DurationElement` is an OSMF-provided `ProxyElement` subclass that adds temporal capabilities to a non-temporal media element. For example, you can use it to wrap a set of `ImageElement` objects to create a slide show, as explained in “[Using DurationElement](#)” on page 20.

Other `ProxyElement` possibilities include:

- Unskippable video ads, implemented by a `ProxyElement` that blocks the `Seek` trait of its wrapped `MediaElement`.
- User analytics, via a `ProxyElement` that listens for changes to the wrapped `MediaElement` and reports them to a server.
- Seamless video switching, from a `ProxyElement` that wraps two `VideoElement` objects, and switches from one to the other without rebuffering.



Nesting ProxyElement objects

Using DurationElement

`DurationElement` is a proxy element that provides temporal capabilities. You can use it by itself to provide delays between the presentation of other media elements in a composition. Or, you can use `DurationElement` to wrap a non-temporal `MediaElement`, such as an `ImageElement`, to give it temporal capabilities.

`DurationElement` overrides the wrapped element's `TimeTrait` to return a custom instance of that trait. `DurationElement` does not work with elements that already have a `TimeTrait`, such as `VideoElement`. See [“Traits”](#) on page 27 for a more detailed discussion of this type.

This first example shows `DurationElement` instances being used to provide delays between video elements:

```
var sequence:SerialElement = new SerialElement();

sequence.addChild (new VideoElement(new URLResource
    ("http://www.example.com/video1.flv")));
sequence.addChild(new DurationElement(5));
sequence.addChild(new VideoElement(new URLResource
    ("http://www.example.com/ad.flv")));
sequence.addChild(new DurationElement(5));
sequence.addChild(new VideoElement(new URLResource
    ("http://www.example.com/video2.flv")));

// Assign the SerialElement to the MediaPlayer.
player.media = sequence;
```

The following example presents a sequence of rotating banners. The delays separating the appearances of the banners are created with `DurationElement` instances. In addition, the images themselves are wrapped in `DurationElement` objects to enable them to support a duration.

```
// The first banner does not appear for five seconds.
// Each banner is shown for 20 seconds.
// There is a 15-second delay between images.

var bannerSequence:SerialElement = new SerialElement();

bannerSequence.addChild(new DurationElement(5));
bannerSequence.addChild(new DurationElement(20,new ImageElement(new URLResource
    ("http://www.example.com/banner1.jpg"))));
bannerSequence.addChild(new DurationElement(15));
bannerSequence.addChild(new DurationElement(20,new ImageElement(new URLResource
    ("http://www.example.com/banner2.jpg"))));
bannerSequence.addChild(new DurationElement(15));
bannerSequence.addChild(new DurationElement(20,new ImageElement(new URLResource
    ("http://www.example.com/banner3.jpg"))));
```

Note: To specify a start and end time for a video element, use the `StreamingURLResource` properties `clipStartTime` and `clipEndTime`.

Metadata

OSMF defines metadata as key-value pairs, where keys are strings, and values are arbitrary objects that can be any ActionScript data structures. The OSMF `Metadata` class provides a strongly typed API for working with these key-value pairs, as well as events for detecting changes to the metadata.

Two areas where OSMF provides defined metadata mechanisms are for layout control and cue points.

Layout control

As described in “[Basic Playback](#)” on page 4, using the `MediaContainer` class (either directly or indirectly via `MediaPlayerSprite`) is what gives your player visibility. `MediaContainer` also provides control over how content is sized and positioned within your media player.

When you construct a media container, you can specify `LayoutMetadata` properties that apply to each `MediaElement` child of the container. You use these properties to position and scale the child media elements. If you are using `MediaPlayerSprite`, you can still set `LayoutMetadata` properties for the child `Sprite` object.

Some important `LayoutMetadata` properties are:

`x`, `y`, `width`, `height` (in pixels): These values are explicit directions on position and size settings. Regardless of the child's context, these values are respected. Since these values are **absolute** (not relative to the parent container's size), it is possible that the items in the layout do not fit. If that happens, then the basic layout clips out the offending child regions.

`percentX`, `percentY`, `percentWidth`, `percentHeight` (as a percentage of the corresponding container's size): These properties express position and size for the child element as percentages of the parent container size. The absolute layout properties (`x`, `y`, `width`, and `height`) take precedence over the **relative** values that this class defines.

`left`, `top`, `right`, `bottom` (in pixels): These properties define an absolute offset for child content from the parent container's borders. Both absolute layout properties and relative layout properties, defined above, take precedence over these **offset** layout properties.

`paddingTop`, `paddingLeft`, `paddingBottom`, `paddingRight` (in pixels): If **padding** `LayoutMetadata` values are defined, other settings do not take precedence over them or override them. Padding is used to create a blank border around content. The padding is applied within the bounds of the current content area, rather than being applied to the outside border of the content. Thus, padding does not change the dimensions of the child element's overall content area. Padding does, however, reduce the area within the bounds of the child element that is available to display content. Note that the right and bottom padding margins are created by altering the width or height of the content. If these values are undefined, then no padding can be applied, even if the top and left padding properties are set.

`snapToPixel`: This option ensures that the layout renderer positions and sizes elements on whole pixel boundaries. For most media, putting it on sub-pixel boundaries results in a less sharp image, therefore pixel clamping is set to `true` by default.

The following shows a simple example of applying the `width` and `height` `LayoutMetadata` properties to the content of a media container.


```
package org.osmf.layout
{
    import flash.display.Sprite;
    import org.osmf.media.MediaPlayerSprite;
    import org.osmf.media.URLResource;

    public class LayoutMetadataExample extends Sprite
    {
        public function LayoutMetadataExample()
        {
            var sprite:MediaPlayerSprite = new MediaPlayerSprite();
            addChild(sprite);
            sprite.resource = new URLResource("http://myserver.com/content/test.flv");

            // Construct a metadata instance and
            // set an absolute width and height of 100 pixels:
            var layoutMetadata:LayoutMetadata = new LayoutMetadata();
            layoutMetadata.width = 100;
            layoutMetadata.height = 100;

            // Apply the layout metadata to the media element at hand, resulting
            // in the video displaying at 100x100 pixels:
            sprite.media.addMetadata(LayoutMetadata.LAYOUT_NAMESPACE, layoutMetadata);
        }
    }
}
```

Timelines and cue points

While you don't have to use OSMF to use cue points, the framework does provide cue point support. The underlying logic for ActionScript cue points in OSMF can be found in `TimelineMetadata` in the `org.osmf.metadata` package.

`TimelineMetadata` encapsulates metadata associated with the timeline of a `MediaElement`. `TimelineMetadata` uses the `TimelineMarker` class to represent both keys and values. That is, a `TimelineMarker` is stored as both key and value. The `TimelineMarker` class represents an individual marker in the timeline of a `MediaElement` object. `TimelineMarker` objects are aggregated by a `TimelineMetadata` object.

A `TimelineMetadata` object dispatches a `TimelineMetadataEvent` when the `currentTime` property of the media element's `TimeTrait` matches any of the time values in its collection of `TimelineMarker` objects.

The `addMarker` method maintains the `TimelineMarker` instances in time order. If another `TimelineMarker` with the same time value exists within this object, then the existing value is overwritten. The AS `TimelineMetadata` cue points are accurate within 250 milliseconds and reliable when the user seeks or pauses.

The `Metadata` subclass `CuePoint` represents a cue point in the timeline of a media element. A cue point is a media time value that has an associated action or piece of information. Typically, cue points are associated with video timelines to represent navigation points or event triggers. The `CuePoint` class extends `TimelineMarker` and can be added to a `TimelineMetadata` object.

You can add custom data via `CuePoint`. If you need more data for a single cue point, you can supply it when you instantiate `CuePoint`. The fourth parameter in the `CuePoint` constructor is an `Object`, which can be anything you want, such as:

```
var cuePoint:CuePoint = new CuePoint(CuePointType.ACTIONSCRIPT, 162.12, "MyCuePoint",
{key1:"value 1", key2:"value 2"});
```

The following sample demonstrates adding an event listener to retrieve embedded cue points for an F4V (H.264) file. These are cue points that were embedded during encoding (that is, at encode time):

```
package
{
    import flash.display.Sprite;
    import flash.display.StageAlign;
    import flash.display.StageScaleMode;
    import org.osmf.elements.VideoElement;
    import org.osmf.events.MediaElementEvent;
    import org.osmf.events.TimelineMetadataEvent;
    import org.osmf.media.MediaPlayerSprite;
    import org.osmf.media.URLResource;
    import org.osmf.metadata.CuePoint;
    import org.osmf.metadata.TimelineMetadata;

    public class TimelineMetadataExample extends Sprite
    {
        public function TimelineMetadataExample()
        {
            super();

            stage.scaleMode = StageScaleMode.NO_SCALE;
            stage.align = StageAlign.TOP_LEFT;

            var mediaPlayerSprite:MediaPlayerSprite = new MediaPlayerSprite();
            var urlResource:URLResource = new URLResource
                ("rtmp://myserver.net/content/test.f4v");
            videoElement= new VideoElement();
            videoElement.resource = urlResource;
            videoElement.addEventListener(MediaElementEvent.METADATA_ADD, onMetadataAdd);
            addChild(mediaPlayerSprite);
            mediaPlayerSprite.media = videoElement;
        }
        private function onMetadataAdd(event:MediaElementEvent):void
        {
            if (event.namespaceURL == CuePoint.DYNAMIC_CUEPOINTS_NAMESPACE)
            {
                var timelineMetadata:TimelineMetadata = videoElement.getMetadata
                    (CuePoint.DYNAMIC_CUEPOINTS_NAMESPACE) as TimelineMetadata;
                timelineMetadata.addEventListener
                    (TimelineMetadataEvent.MARKER_TIME_REACHED, onCuePoint);
            }
        }
        private function onCuePoint(event:TimelineMetadataEvent):void
        {
            var cuePoint:CuePoint = event.marker as CuePoint;
            trace("Cue Point at " + cuePoint.time);
        }
        private var videoElement:VideoElement;
    }
}
```

You can also set runtime ActionScript cue points, as shown here:

```
var dynamicTimelineMetadata = new TimelineMetadata(mediaPlayer.media);
mediaPlayer.media.addMetadata
    (CuePoint.DYNAMIC_CUEPOINTS_NAMESPACE, dynamicTimelineMetadata);

// You can add as many cue points as you like here
var cuePoint:CuePoint = new CuePoint
    (CuePointType.ACTIONSCRIPT, 120/*this is a time value in seconds*/, "my cue point",
        null/*this could be custom name-value pair data*/);

dynamicTimelineMetadata.addMarker(cuePoint);
```

Chapter 6: Advanced Topics

Open Source Media Framework provides the following advanced tools for your player development and customization:

- For debugging and optimizing your code, OSMF supplies detailed log messages. And because OSMF logging data is produced in a standard format, it is shareable between a player and plug-ins. For details, see “[Logging](#)” on page 25.
- OSMF media players use the `MediaElement` class to work with content. But, to really understand `MediaElement` or to go beyond it, you need to know about “[Traits](#)” on page 27.
- When you embed an OSMF media player in an HTML page, communication between the player and the HTML elements of the page can be necessary. OSMF allows your player to interact with JavaScript, as described in “[External HTML control with JavaScript](#)” on page 28.
- Third-party, plug-in services (such as advertising, analytics, or content delivery networks) are widely available to enhance the services that your media player offers. See “[Plug-ins](#)” on page 30 for a discussion of how to use these services.

Logging

OSMF logging data is produced in a standard format, making it shareable between a player and plug-ins. The default implementation sends log messages to the debug console of the Flash Builder and Flex Builder development environments.

By default, logging is turned off. To turn it on, set the compiler option `-define CONFIG::LOGGING` to `true`.

The `Logger` class defines the capabilities of a logger, the object that OSMF applications interact with to write logging messages. Usually in an OSMF application there are multiple instances of `Logger`.

The `LoggerFactory` class generates `Logger` objects and serves as the initial contact point for an application. There is typically only one instance of `LoggerFactory` per application. If you want to enhance the logging abilities of your OSMF application, you can subclass the `LoggerFactory` class.

This first code sample provides a brief outline of creating a custom `Logger` subclass:

```
package
{
    import org.osmf.logging.Logger;

    public class ExampleLogger extends Logger
    {
        public function ExampleLogger(category:String)
        {
            super(category);
        }

        override public function debug(message:String, ... rest):void
        {
            trace(message);
        }
    }
}
```

This next sample code creates a new `LoggerFactory` subclass which uses the custom `Logger` subclass:

```
package
{
    import org.osmf.logging.Logger;
    import org.osmf.logging.LoggerFactory;

    public class ExampleLoggerFactory extends LoggerFactory
    {
        public function ExampleLoggerFactory()
        {
            super();
        }
        override public function getLogger(category:String):Logger
        {
            return new ExampleLogger(category);
        }
    }
}
```

The final sample shows the application calling the custom `LoggerFactory`:

```
package
{
    import flash.display.Sprite;
    import flash.display.StageAlign;
    import flash.display.StageScaleMode;
    import org.osmf.elements.VideoElement;
    import org.osmf.logging.Logger;
    import org.osmf.logging.Log;
    import org.osmf.media.MediaPlayerSprite;
    import org.osmf.media.URLResource;

    public class LoggerSample extends Sprite
    {
        public function LoggerSample()
        {
            super();

            Log.loggerFactory = new ExampleLoggerFactory();
            logger = Log.getLogger("LoggerSample");

            stage.scaleMode = StageScaleMode.NO_SCALE;
            stage.align = StageAlign.TOP_LEFT;

            var mediaPlayerSprite:MediaPlayerSprite = new MediaPlayerSprite();
            var urlResource:URLResource = new URLResource
                ("rtmp://myserver.net/content/test_640_500_short");
            var videoElement:VideoElement = new VideoElement(urlResource);

            addChild(mediaPlayerSprite);

            logger.debug("Ready to play video at " + urlResource.url);
            mediaPlayerSprite.media = videoElement;
        }
        private var logger:Logger;
    }
}
```

For a robust example of logging with OSMF, see [the Strobe Media Playback debug console](#). For further information on Strobe Media Playback's implementation of OSMF logging, see [the debug console's wiki page](#).

Traits

“[Media elements and resources](#)” on page 4 introduced the ways OSMF represents content via the `MediaElement` class and its subclasses. However, if you want to perform more in-depth work with media elements, it can be useful to understand the `MediaElement` type in detail.

`MediaElement` objects express what they do via traits. Traits represent fundamental capabilities or characteristics of a piece of media. A trait can represent the availability of audio, or the ability to play or seek. The sum of all traits on a media element define the overall abilities of the media element. Traits are dynamic, and are added and removed from media at runtime, in response to other events.


Different media elements combine different sets of traits, depending on the characteristics of the media they represent. For example, an audio element may have the `audio`, `load`, `play`, and `time` traits. You can use traits to build custom `MediaElement` subclasses.

The total set of traits in the framework is fixed, as they form the core vocabulary of the OSMF system. Creating custom individual traits is not supported. The following table lists a few basic traits and some of the media types to which they apply.

MediaElement	PlayTrait?	DisplayObjectTrait?	TimeTrait?	AudioTrait?
VideoElement	Yes	Yes	Yes	Yes
AudioElement	Yes	No	Yes	Yes
ImageElement	No	Yes	No	No

`MediaPlayer` handles the traits of any `MediaElement` subclass. For example, `MediaPlayer` inspects the traits of a media element before loading it, in order to properly set up the player.

Trait values are subject to change, and traits dispatch events to signal these changes. For example, if the user changes the volume level, a `volumeChange` event is dispatched, because the audio trait's `volume` property has changed. `MediaPlayer` constantly monitors the availability and state of the media element's traits and responds to trait change events.



Visualizing traits

Traits typically expose getters without setters, except for traits that require externally settable properties, such as the volume or pan properties of an audio trait. You find out whether a media element has a trait of a given type by calling the `MediaElement.hasTrait()` method. For example, if `hasTrait(MediaTraitType.AUDIO)` returns `true`, you know that the element has an audio trait. You use `MediaElement.getTrait(MediaTraitType.AUDIO)` to get an object of this type.

External HTML control with JavaScript

Say that a web page contains both a video in an OSMF player and an HTML banner ad at the top of the page. When the user clicks the banner ad, the banner ad grows to cover the entire page, and the video in the OSMF player pauses. When the user dismisses the ad, the video resumes playing. Such functionality can be implemented using the `HTMLMediaContainer` and `HTMLDivElement` classes.

`HTMLMediaContainer` uses the Flash external interface feature to expose the container's child media elements to JavaScript. On being initialized, the `HTMLMediaContainer` object inserts JavaScript into the hosting HTML document, using `flash.external.ExternalInterface`. The added JavaScript allows elements in JavaScript to be recognized as `MediaElement` objects by the media player. Next, `HTMLMediaContainer` invokes a callback method that the HTML client can use to do further setup work.

`HTMLElement` is a media element that represents a piece of media external to the Flash SWF and within an HTML region. It serves as a bridge between the OSMF APIs for controlling media, and a corresponding (external) JavaScript implementation. The `HTMLElement` class is used to address specific portions of the hosting HTML page from the media player.

For example, `HTMLElement` can be used to wrap an HTML banner ad:

```
var banner1:HTMLElement = new HTMLElement();  
banner1.resource = new URLResource("http://www.iab.net/media/image/468x60.gif");
```

`HTMLMediaContainer` invokes a callback method in JavaScript when it gets an `HTMLElement` object added (which in turn happens when `HTMLElement` is added to the `HTMLMediaContainer` instance). The callback invoked on an element being added to `HTMLMediaContainer` includes a reference to a `MediaElement` JavaScript object.

This JavaScript object holds `boolean` fields that represent whether the element is playable, temporal, and audible. These switches can be set from JavaScript at will. On doing so, for OSMF, the element appears to respectively carry the `PlayTrait`, `TimeTrait`, and `AudioTrait` traits.

All `HTMLElement` objects also have the trait `LoadTrait`. On the element receiving a 'load' or 'unload' instruction, it forwards the command to JavaScript. The JavaScript `MediaElement` allows for the loaded state to be set. When no 'load' or 'unload' methods have been set for the element on the JavaScript side, the loaded state is governed by the Flash side.

When a trait is switched on from JavaScript this results in its corresponding trait being added to the media element on the OSMF side. On the JavaScript side, an API allowing the trait to operate is made available.

The following brief sample shows one way to handle this interaction. For an extended ActionScript and HTML sample, see [HTMLMediaContainerSample](#).


```
package
{
    import flash.display.Sprite;
    import org.osmf.elements.HTMLDivElement;
    import org.osmf.media.URLResource;
    public class HTMLMediaContainerExample extends Sprite
    {
        public function HTMLMediaContainerExample()
        {
            super();

            // This invokes a JavaScript callback (onHTMLMediaContainerConstructed)
            // that adds listeners to new elements being added to the container.
            var container:HTMLMediaContainer = new HTMLMediaContainer();

            var element:HTMLDivElement = new HTMLDivElement();
            element.resource = new URLResource("http://example.com/asset/handle.pgn");

            // This invokes a JavaScript callback (container.onElementAdd) that
            // allows JavaScript to process the passed URL, and to communicate back
            // to the framework what traits the element supports:
            container.addMediaElement(element);
        }
    }
}
```

Plug-ins

A modern media player does much more than play media. It can also use a content delivery network (CDN), present advertising, capture user events to report to an analytics server, and so on. But the media player does not usually handle this work by itself. This additional functionality is typically provided by third-party software known as “plug-ins.”

A plug-in is code that you invite to work with your player. When you load a plug-in, you give it permission to provide additional functionality for your player. Plug-ins are not given unlimited access to your media player. OSMF acts as a broker between your media player and the plug-in. This approach ensures that communication between the media player and the plug-in is standardized, making it simple to add, update, or switch plug-ins.

Plug-ins can define additional behavior for media, a new type of media, or both. Plug-ins provide the following benefits:

- Enable additional or customized functionality without requiring changes to the underlying media player code.
- Simplify integration with third-party solutions and services. Examples include advertising, reporting, or content delivery.
- Allow implementation of complementary social media features such as rating and sharing.
- Can be static or dynamic. Static plug-ins are incorporated into the media player at compile time. Dynamic plug-ins are loaded by the media player at runtime from a SWF file.

To see some examples of plug-ins, check out the [OSMF plug-in partner page](#).

Using plug-ins

To use plug-ins, your media player must be written using `DefaultMediaFactory` or another `MediaFactory` subclass, rather than instantiating media elements directly. If there are plug-ins available to your player, OSMF uses `MediaFactory` to interact with them. `MediaFactory` finds the appropriate plug-in for the media resource type you provide. Then, in conjunction with the plug-in, `MediaFactory` returns a plug-in-enabled `MediaElement` to your player.

On the plug-in side, each plug-in is required to provide a media description of type `MediaFactoryItem`. The `MediaFactoryItem` object has a `canHandleResourceFunction` property that identifies the resource types that the plug-in can handle.

Static plug-ins

To begin using a static plug-in, download the posted plug-in source or SWC library. Add the result to your media player project. The advantage of using static plug-ins is that there is no load time for the plug-in, since it's compiled directly into the application. After you load a plug-in, there is no further direct interaction between your player code and the plug-in. You provide a media resource to the factory, and the plug-in alters the resulting `MediaElement` based on its capabilities.

- 1 Link the library or source of the plug-in.
- 2 Create a `PluginInfoResource` object from an instance of the `PluginInfo` subclass:

```
var myPlugin:PluginInfoResource = new PluginInfoResource(new MyPluginInfo);
```

Dynamic plug-ins

With a dynamic plug-in, you can use the plug-in from its original server location, or you can download and post the plug-in on a server you control. If you use the original location, then when the developer posts updates to the plug-in your player is automatically updated, too. Either way, with dynamic plug-ins you do not need to recompile the player application to include updates.

When it loads a plug-in dynamically, OSMF verifies that the plug-in was compiled with a version of OSMF that the player supports. If there is a version mis-match between the player and the plug-in, the plug-in loading fails. Because the player receives only a general plug-in load error, developers may wish to check the version of the plug-in to find out whether this was the cause. The simplest way to check plug-in version compatibility is to call the `PluginInfo.isFrameworkVersionSupported` method in `org.osmf.media`.

The basic steps to load a dynamic plug-in are as follows:

- 1 Create a `URLResource` object that points to the plug-in SWF file location.
- 2 To perform the actual loading:

```
var factory:MediaFactory = new DefaultMediaFactory();  
var myPluginURL:URLResource = new URLResource(PLUGIN_URL_ONMYSERVER);  
factory.loadPlugin(myPluginURL);
```

- 3 Listen for the `MediaFactoryEvent.PLUGIN_LOAD` or `PLUGIN_LOAD_ERROR` events.

```
factory.addEventListener(MediaFactoryEvent.PLUGIN_LOAD, onPluginLoaded);  
factory.addEventListener(MediaFactoryEvent.PLUGIN_LOAD_ERROR, onPluginLoadError);
```

Writing a control bar plug-in

“[Interaction](#)” on page 10 mentions that one way to implement controls for your player is as a plug-in. This section provides an example of one such plug-in. It is not intended to be a complete description of writing plug-ins in general. For that information, see [the OSMF Plug-In Developer's Guide](#).

In the case of this plug-in, the interface object is modeled as a type of media element. For example, you can think of an overlay play button as a piece of media that “plays” in parallel with a video that it controls. This example implements a plug-in that defines a `MediaElement`-based `ControlBarElement` object. The `ControlBarElement` object displays a control bar. The plug-in also contains a reference to the video element it controls.

ControlBarPlugin

The control bar plug-in is implemented as two classes: `ControlBarPlugin` and `ControlBarElement`.

Note: The source for this sample is located at the [OSMF open source repository](#).

The first class, `ControlBarPlugin`, extends the `Sprite` class and is the plug-in SWF document class. It defines the mandatory `pluginInfo` getter method that the framework uses to obtain a `PluginInfo` reference.

It also defines two callback methods. The first, `canHandleResourceCallback`, is invoked when the framework determines whether a plug-in can construct a media element for a given resource type. The second, `mediaElementCreationCallback`, is invoked when the framework is requesting the plug-in to provide a media element instance. The latter returns an object that is defined in the plug-in's other class, “[ControlBarElement](#)” on page 34.

```
package
{
    import __AS3__.vec.Vector;

    import flash.display.Sprite;
    import flash.system.Security;

    import org.osmf.media.MediaElement;
    import org.osmf.media.MediaFactoryItem;
    import org.osmf.media.MediaResourceBase;
    import org.osmf.media.PluginInfo;
    import org.osmf.metadata.Metadata;

    public class ControlBarPlugin extends Sprite
    {
        /**
         * Constructor
         */
        public function ControlBarPlugin()
        {
            // Allow any SWF that loads this SWF to access objects and
            // variables in this SWF.
            Security.allowDomain("*");

            super();
        }

        /**
         * Gives the player the PluginInfo.
         */
        public function get pluginInfo():PluginInfo
        {
            if (_pluginInfo == null)
            {
                var item:MediaFactoryItem
                    = new MediaFactoryItem
                        ( ID
```

```
        , canHandleResourceCallback
        , mediaElementCreationCallback
    );

    var items:Vector.<MediaFactoryItem> = new Vector.<MediaFactoryItem>();
    items.push(item);

    _pluginInfo = new PluginInfo(items, mediaElementCreationNotificationCallback);
}

return _pluginInfo;
}

// Internals
//

public static const ID:String = "com.example.samples.controlbar";
public static const NS_CONTROL_BAR_SETTINGS:String =
"http://www.osmf.org/samples/controlbar/settings";
public static const NS_CONTROL_BAR_TARGET:String =
"http://www.osmf.org/samples/controlbar/target";

private var _pluginInfo:PluginInfo;
private var controlBarElement:ControlBarElement;
private var targetElement:MediaElement;

private function canHandleResourceCallback(resource:MediaResourceBase):Boolean
{
    var result:Boolean;

    if (resource != null)
    {
        var settings:Metadata
            = resource.getMetadataValue(NS_CONTROL_BAR_SETTINGS) as Metadata;

        result = settings != null;
    }

    return result;
}

private function mediaElementCreationCallback():MediaElement
{
    controlBarElement = new ControlBarElement();

    updateControls();

    return controlBarElement;
}
```

```
private function mediaElementCreationNotificationCallback(target:MediaElement):void
{
    // If the control bar has been created,
    // notify it about the just-created element.
    targetElement = target;

    updateControls();
}

private function updateControls():void
{
    if (controlBarElement != null && targetElement != null && controlBarElement !=
targetElement)
    {
        controlBarElement.addReference(targetElement);
    }
}
}
```

ControlBarElement

The `ControlBarElement` class defines the control bar object, based on `MediaElement`. It exposes the control bar's Sprite by adding a `DisplayObjectTrait` to the media element.

```
package
{
    import org.osmf.chrome.assets.AssetsManager;
    import org.osmf.chrome.configuration.LayoutAttributesParser;
    import org.osmf.chrome.configuration.WidgetsParser;
    import org.osmf.chrome.widgets.Widget;
    import org.osmf.layout.LayoutMetadata;
    import org.osmf.media.MediaElement;
    import org.osmf.media.MediaResourceBase;
    import org.osmf.metadata.Metadata;
    import org.osmf.traits.DisplayObjectTrait;
    import org.osmf.traits.MediaTraitType;

    public class ControlBarElement extends MediaElement
    {
        // Embedded assets (see configuration.xml for their assignments):
        //

        [Embed(source="../../assets/configuration.xml", mimeType="application/octet-stream")]
        private static const CONFIGURATION_XML:Class;

        [Embed(source="../../assets/Standard0755.swf#Standard0755")]
        private static const DEFAULT_FONT:Class;

        [Embed(source="../../assets/backDrop.png")]
        private static const BACKDROP:Class;

        [Embed(source="../../assets/pause_disabled.png")]
        private static const PAUSE_DISABLED:Class;
        [Embed(source="../../assets/pause_up.png")]
        private static const PAUSE_UP:Class;
    }
}
```

```
[Embed(source="../../assets/pause_down.png")]
private static const PAUSE_DOWN:Class;

[Embed(source="../../assets/stop_disabled.png")]
private static const STOP_DISABLED:Class;
[Embed(source="../../assets/stop_up.png")]
private static const STOP_UP:Class;
[Embed(source="../../assets/stop_down.png")]
private static const STOP_DOWN:Class;

[Embed(source="../../assets/play_disabled.png")]
private static const PLAY_DISABLED:Class;
[Embed(source="../../assets/play_up.png")]
private static const PLAY_UP:Class;
[Embed(source="../../assets/play_down.png")]
private static const PLAY_DOWN:Class;

[Embed(source="../../assets/scrubber_disabled.png")]
private static const SCRUBBER_DISABLED:Class;
[Embed(source="../../assets/scrubber_up.png")]
private static const SCRUBBER_UP:Class;
[Embed(source="../../assets/scrubber_down.png")]
private static const SCRUBBER_DOWN:Class;
[Embed(source="../../assets/scrubBarTrack.png")]
private static const SCRUB_BAR_TRACK:Class;

// Public interface
//

public function addReference(target:MediaElement):void
{
    if (this.target == null)
    {
        this.target = target;

        processTarget();
    }
}

private function processTarget():void
{
    if (target != null && settings != null)
    {
        // We use the NS_CONTROL_BAR_TARGET namespaced metadata in order
        // to find out if the instantiated element is the element that our
        // control bar should control:
        var targetMetadata:Metadata =
target.getMetadata(ControlBarPlugin.NS_CONTROL_BAR_TARGET);
        if (targetMetadata)
        {
            if (    targetMetadata.getValue(ID) != null
                && targetMetadata.getValue(ID) == settings.getValue(ID)
            )
            {
                controlBar.media = target;
            }
        }
    }
}
```

```

    }
}

// Overrides
//

override public function set_resource(value:MediaResourceBase):void
{
    // Right after the media factory has instantiated us, it will set the
    // resource that it used to do so. We look the NS_CONTROL_BAR_SETTINGS
    // namespaced metadata, and retain it as our settings record
    // (containing only one field: "ID" that tells us the ID of the media
    // element that we should be controlling):
    if (value != null)
    {
        settings
            = value.getMetadataValue(ControlBarPlugin.NS_CONTROL_BAR_SETTINGS) as
Metadata;

        processTarget();
    }

    super.resource = value;
}

override protected function setupTraits():void
{
    // Set up a control bar using the ChromeLibrary:
    setupControlBar();

    // Use the control bar's layout metadata as the element's layout metadata:
    var layoutMetadata:LayoutMetadata = new LayoutMetadata();
    LayoutAttributesParser.parse(controlBar.configuration, layoutMetadata);
    addMetadata(LayoutMetadata.LAYOUT_NAMESPACE, layoutMetadata);

    // Signal that this media element is viewable: create a DisplayObjectTrait.
    // Assign controlBar (which is a Sprite) to be our view's displayObject.
    // Additionally, use its current width and height for the trait's mediaWidth
    // and mediaHeight properties:
    viewable = new DisplayObjectTrait(controlBar, controlBar.measuredWidth,
controlBar.measuredHeight);
    // Add the trait:
    addTrait(MediaTraitType.DISPLAY_OBJECT, viewable);

    controlBar.measure();

    super.setupTraits();
}

// Internals
//

private function setupControlBar():void
{
    try
    {
        var configuration:XML = XML(new CONFIGURATION_XML());
    }
}

```

```

        var assetsManager:AssetsManager = new AssetsManager();
        assetsManager.addConfigurationAssets(configuration);
        assetsManager.load();

        var widgetsParser:WidgetsParser = new WidgetsParser()
        widgetsParser.parse(configuration.widgets.*, assetsManager);

        controlBar = widgetsParser.getWidget("controlBar");
    }
    catch (error:Error)
    {
        trace("WARNING: failed setting up control bar:", error.message);
    }
}

private var settings:Metadata;

private var target:MediaElement;
private var controlBar:Widget;
private var viewable:DisplayObjectTrait;

/* static */

private static const ID:String = "ID";
}
}

```

ControlBarPluginSample

The OSMF player sample application consists of two classes. The first contains much of the standard code that is required in setting up any OSMF player application, and is omitted here. It's available from the [OSMF public code repository](#).

The second class, called `ControlBarPluginSample`, is the main tester application and is located with the [OSMF sample plugins](#).

Using `OSMFConfiguration`, the `MediaFactory.loadPlugin` method loads `ControlBarPlugin.swf`. Meanwhile, a parallel media element is set up that acts as the root of the aggregate media that is played back. The first child is then added, a video that plays an OSMF logo animation. Once the plug-in is loaded, the second element is added: a control bar media element, instantiated by the code on `ControlBarPlugin.swf`. This application uses `ControlBarPlugin` as a static plug-in for ease of debugging.

When the plug-in loaded, it registers itself with the media factory. When the factory is asked to create an element for the given resource, it goes over all types and plug-ins that registered with it. Ultimately, this results in the `canHandleResourceCallback` method in `ControlBarPlugin.as` being invoked. The implementation of that method returns `true` if it finds the resource to have metadata that goes by the namespace defined in `NS_CONTROL_BAR_SETTINGS`. When this method returns `true`, the framework asks the plug-in to instantiate a media element.

When the media element is instantiated, the plug-in is informed through the `mediaElementCreationNotificationFunction`. The control bar plug-in's implementation of this function calls `addReference` to associate the created element with the control bar. But it only does so if the element has metadata with a namespace URL that matches the namespace defined in `NS_CONTROL_BAR_TARGET`.

The last step in matching the control bar to the element it controls takes place inside the control bar element's `addReference` method. If there is an element with the same ID as the element's own ID from `NS_CONTROL_BAR_SETTINGS`, the element is assigned as the control bar's target element.

```
package
{
    import flash.display.Sprite;

    import org.osmf.elements.ParallelElement;
    import org.osmf.events.MediaFactoryEvent;
    import org.osmf.layout.HorizontalAlign;
    import org.osmf.layout.LayoutMetadata;
    import org.osmf.layout.VerticalAlign;
    import org.osmf.media.*;
    import org.osmf.metadata.Metadata;

    [SWF(width="640", height="360", backgroundColor="0x000000", frameRate="25")]
    public class ControlBarPluginSample extends Sprite
    {
        public function ControlBarPluginSample()
        {
            // Construct an OSMFConfiguration helper class:
            osmf = new OSMFConfiguration();

            // Construct the main element to play back. This is a
            // parallel element, that holds the main content to
            // play back, and the control bar (from a plug-in) as its
            // children:
            osmf.mediaElement = constructRootElement();
            osmf.view = this;

            // Add event listeners to the plug-in manager to receive
            // a heads-up when the control bar plug-in finishes loading:
            osmf.factory.addEventListener(MediaFactoryEvent.PLUGIN_LOAD, onPluginLoaded);
            osmf.factory.addEventListener(MediaFactoryEvent.PLUGIN_LOAD_ERROR,
onPluginLoadError);

            // Ask the plug-in manager to load the control bar plug-in:
            osmf.factory.loadPlugin(pluginResource);
        }

        // Internals
        //

        private var osmf:OSMFConfiguration;
        private var rootElement:ParallelElement;

        private function onPluginLoaded(event:MediaFactoryEvent):void
        {
            // The plugin loaded successfully. We can now construct a control
            // bar media element, and add it as a child to the root parallel element:
            rootElement.addChild(constructControlBarElement());
        }

        private function onPluginLoadError(event:MediaFactoryEvent):void
        {
            trace("ERROR: the control bar plugin failed to load.");
        }
    }
}
```

```

    }

    private function constructRootElement():MediaElement
    {
        // Construct a parallel media element to hold the main content,
        // and later on, the control bar.
        rootElement = new ParallelElement();
        rootElement.addChild(constructVideoElement());

        // Use the layout api to set the parallel element's width and
        // height. Make it the same size as the stage currently is:
        var rootElementLayout:LayoutMetadata = new LayoutMetadata();
        rootElement.addMetadata(LayoutMetadata.LAYOUT_NAMESPACE, rootElementLayout);
        rootElementLayout.width = stage.stageWidth;
        rootElementLayout.height = stage.stageHeight;

        return rootElement;
    }

    private function constructVideoElement():MediaElement
    {
        // Construct a metadata object that we can append to the video's collection
        // of metadata. The control bar plug-in uses the metadata to identify
        // the video element as its target:
        var controlBarTarget:Metadata = new Metadata();
        controlBarTarget.addValue(ID, "mainContent");

        // Construct a video element:
        var video:MediaElement = osmf.factory.createMediaElement(new
URLResource(VIDEO_URL));

        // Add the metadata to the video's metadata:
        video.addMetadata(ControlBarPlugin.NS_CONTROL_BAR_TARGET, controlBarTarget);

        return video;
    }

    private function constructControlBarElement():MediaElement
    {
        // Construct a metadata object to send to the media factory on
        // requesting a control bar element to be instantiated. The factory
        // uses it to parameterize the element. Specifically, the ID field
        // tells the plug-in what the ID of the content it should control is:
        var controlBarSettings:Metadata = new Metadata();
        controlBarSettings.addValue(ID, "mainContent");

        // Add the metadata to an otherwise empty media resource object:
        var resource:MediaResourceBase = new MediaResourceBase();
        resource.addMetadataValue(ControlBarPlugin.NS_CONTROL_BAR_SETTINGS,
controlBarSettings);

        // Request the media factory to construct a control bar element. The
        // factory knows a control bar element is requested by inspecting
        // the resource's metadata (and encountering a metadata object of namespace
        // NS_CONTROL_BAR_SETTINGS there):
        var controlBar:MediaElement = osmf.factory.createMediaElement(resource);
    }

```

```
        // Set some layout properties on the control bar. Specifically, have it
        // appear at the bottom of the parallel element, horizontally centered:
        var layout:LayoutMetadata = controlBar.getMetadata(LayoutMetadata.LAYOUT_NAMESPACE)
as LayoutMetadata;
        if (layout == null)
        {
            layout = new LayoutMetadata();
            controlBar.addMetadata(LayoutMetadata.LAYOUT_NAMESPACE, layout);
        }
        layout.verticalAlign = VerticalAlign.BOTTOM;
        layout.horizontalAlign = HorizontalAlign.CENTER;

        // Make sure that the element shows over the video: elements with a
        // higher-order number set are placed higher in the display list:
        layout.index = 1;

        return controlBar;
    }

    /* static */

    private static const VIDEO_URL:String
        = "http://mediapm.edgesuite.net/osmf/content/test/logo_animated.flv";

    private static var ID:String = "ID";

    // Comment out to load the plug-in for a SWF (instead of using static linking, for testing):
    //private static const pluginResource:URLResource = new
URLResource("http://mediapm.edgesuite.net/osmf/swf/ControlBarPlugin.swf");

    private static const pluginResource:PluginInfoResource = new PluginInfoResource(new
ControlBarPlugin().pluginInfo);
    }
}
```